SM_Kerr_Newman

September 22, 2020

1 Kerr-Newman spacetime

This notebook demonstrates a few capabilities of SageMath in computations regarding Kerr-Newman spacetime. The corresponding tools have been developed within the SageManifolds project.

Click here to download the notebook file (ipynb format). To run it, you must start SageMath within the Jupyter notebook, via the command `sage -n jupyter`

NB: a version of SageMath at least equal to 9.1 is required to run this notebook:

[1]:
```
version()
```

[1]: 'SageMath version 9.2.beta13, Release Date: 2020-09-21'

First we set up the notebook to display mathematical objects using LaTeX rendering:

[2]:
```
%display latex
```

and we initialize a time counter for benchmarking:

[3]:
```
import time
comput_time0 = time.perf_counter()
```

Since some computations are quite heavy, we ask for running them in parallel on 8 threads:

[4]:
```
Parallelism().set(nproc=8)
```

1.1 Spacetime manifold

We declare the Kerr-Newman spacetime (or more precisely the part of it covered by Boyer-Lindquist coordinates) as a 4-dimensional Lorentzian manifold $\mathcal{M}$:

[5]:
```
M = Manifold(4, 'M', latex_name=r'\mathcal{M}', structure='Lorentzian')
```

We then introduce the standard Boyer-Lindquist coordinates as a chart `BL` (for *Boyer-Lindquist*) on $\mathcal{M}$, via the method `chart()`, the argument of which is a string (delimited by `r"..."` because of the backslash symbols) expressing the coordinates names, their ranges (the default is $(-\infty, +\infty)$) and their LaTeX symbols:

[6]:
```
BL.<t,r,th,ph> = M.chart(r't r th:(0,pi):\theta ph:(0,2*pi):\phi')
print(BL); BL
```

Chart (M, (t, r, th, ph))

[6]:

$(\mathcal{M},(t,r,\theta,\phi))$

Metric tensor

The 3 parameters m, a and q of the Kerr-Newman spacetime are declared as symbolic variables:

[7]:
```
var('m a q')
```

[7]:

(m,a,q)

We get the (yet undefined) spacetime metric:

[8]:
```
g = M.metric()
```

The metric is defined by its components in the coordinate frame associated with Boyer-Lindquist coordinates, which is the current manifold's default frame:

[9]:
```
rho2 = r^2 + (a*cos(th))^2
Delta = r^2 -2*m*r + a^2 + q^2
g[0,0] = -1 + (2*m*r-q^2)/rho2
g[0,3] = -a*sin(th)^2*(2*m*r-q^2)/rho2
g[1,1], g[2,2] = rho2/Delta, rho2
g[3,3] = (r^2 + a^2 + (2*m*r-q^2)*(a*sin(th))^2/rho2)*sin(th)^2
g.display()
```

[9]:

$$g = \left(-\frac{q^2-2\,mr}{a^2\cos(\theta)^2+r^2}-1\right)\mathrm{d}t\otimes\mathrm{d}t + \left(\frac{\left(q^2-2\,mr\right)a\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right)\mathrm{d}t\otimes\mathrm{d}\phi + \left(\frac{a^2\cos(\theta)^2+r^2}{a^2+q^2-2\,mr+r^2}\right)\mathrm{d}r\otimes\mathrm{d}r + \left(a^2\cos(\theta)^2+r^2\right)\mathrm{d}\theta\otimes\mathrm{d}\theta + \left(\frac{\left(q^2-2\,mr\right)a\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right)\mathrm{d}\phi\otimes\mathrm{d}t - \left(\frac{\left(q^2-2\,mr\right)a^2\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}-a^2-r^2\right)\sin(\theta)^2\,\mathrm{d}\phi\otimes\mathrm{d}\phi$$

The list of the non-vanishing components:

[10]:
```
g.display_comp()
```

[10]:

$$\begin{array}{lcl} g_{t\,t} & = & -\frac{q^2-2\,mr}{a^2\cos(\theta)^2+r^2}-1 \\ g_{t\,\phi} & = & \frac{\left(q^2-2\,mr\right)a\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} \\ g_{r\,r} & = & \frac{a^2\cos(\theta)^2+r^2}{a^2+q^2-2\,mr+r^2} \\ g_{\theta\,\theta} & = & a^2\cos(\theta)^2+r^2 \\ g_{\phi\,t} & = & \frac{\left(q^2-2\,mr\right)a\sin(\theta)^2}{a^2\cos(\theta)^2+r^2} \\ g_{\phi\,\phi} & = & -\left(\frac{\left(q^2-2\,mr\right)a^2\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}-a^2-r^2\right)\sin(\theta)^2 \end{array}$$

The component g^{tt} of the inverse metric:

[11]:
```
g.inverse()[0,0]
```

[11]:

$$\frac{a^4+2\,a^2r^2+r^4-\left(a^4+a^2q^2-2\,a^2mr+a^2r^2\right)\sin(\theta)^2}{2\,mr^3-r^4-(a^2+q^2)r^2-(a^4+a^2q^2-2\,a^2mr+a^2r^2)\cos(\theta)^2}$$

The lapse function:

[12]:
```
N = 1/sqrt(-(g.inverse()[[0,0]])); N
```

[12]:

Scalar field on the 4-dimensional Lorentzian manifold M

[13]:
```
N.display()
```

[13]:

$$\begin{array}{llcl} & \mathcal{M} & \longrightarrow & \mathbb{R} \\ & (t,r,\theta,\phi) & \longmapsto & \frac{\sqrt{a^2\cos(\theta)^2+r^2}\sqrt{|a^2+q^2-2\,mr+r^2|}}{\sqrt{a^4+2\,a^2r^2+r^4-(a^4+a^2q^2-2\,a^2mr+a^2r^2)\sin(\theta)^2}} \end{array}$$

Electromagnetic field tensor

Let us first introduce the 1-form basis associated with Boyer-Lindquist coordinates:

[14]:
```
dBL = BL.coframe(); dBL
```

[14]:

$$(\mathcal{M},(\mathrm{d}t,\mathrm{d}r,\mathrm{d}\theta,\mathrm{d}\phi))$$

The electromagnetic field tensor F is formed as [cf. e.g. Eq. (33.5) of Misner, Thorne & Wheeler (1973)]

[15]:
```
F = q/rho2^2 * (r^2-a^2*cos(th)^2)* dBL[1].wedge( dBL[0] - a*sin(th)^2* dBL[3]␣
↪) + \
    2*q/rho2^2 * a*r*cos(th)*sin(th)* dBL[2].wedge( (r^2+a^2)* dBL[3] - a*␣
↪dBL[0] )
F.set_name('F')
F.display()
```

[15]:

$$F = \left(\frac{a^2q\cos(\theta)^2-qr^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}t\wedge\mathrm{d}r+\left(\frac{2\,a^2qr\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}t\wedge\mathrm{d}\theta+\left(\frac{\left(a^3q\cos(\theta)^2-aqr^2\right)\sin(\theta)^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}r\wedge\mathrm{d}\phi+\left(\frac{2\left(a^3qr+aqr^3\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\theta\wedge\mathrm{d}\phi$$

The list of non-vanishing components:

[16]:
```
F.display_comp()
```

[16]:

$$
\begin{array}{lcl}
F_{t\,r} & = & \frac{a^2 q \cos(\theta)^2 - qr^2}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{t\,\theta} & = & \frac{2\,a^2 qr \cos(\theta)\sin(\theta)}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{r\,t} & = & -\frac{a^2 q \cos(\theta)^2 - qr^2}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{r\,\phi} & = & \frac{\left(a^3 q \cos(\theta)^2 - aqr^2\right)\sin(\theta)^2}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{\theta\,t} & = & -\frac{2\,a^2 qr \cos(\theta)\sin(\theta)}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{\theta\,\phi} & = & \frac{2\left(a^3 qr + aqr^3\right)\cos(\theta)\sin(\theta)}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{\phi\,r} & = & -\frac{\left(a^3 q \cos(\theta)^2 - aqr^2\right)\sin(\theta)^2}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4} \\
F_{\phi\,\theta} & = & -\frac{2\left(a^3 qr + aqr^3\right)\cos(\theta)\sin(\theta)}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4}
\end{array}
$$

The Hodge dual of F:

[17]:
```
star_F = F.hodge_dual(g)
star_F.display()
```

[17]:

$$\star F = \left(\frac{2\,aqr\sqrt{\cos(\theta)+1}\sqrt{-\cos(\theta)+1}\cos(\theta)}{\left(a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4\right)\sin(\theta)}\right) \mathrm{d}t \wedge \mathrm{d}r + \left(-\frac{\left(a^3 q \cos(\theta)^2 - aqr^2\right)\sqrt{\cos(\theta)+1}\sqrt{-\cos(\theta)+1}}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4}\right) \mathrm{d}t \wedge \mathrm{d}\theta + \left(\frac{2\,a^2 qr\sqrt{\cos(\theta)+1}\sqrt{-\cos(\theta)+1}\cos(\theta)\sin(\theta)}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4}\right) \mathrm{d}r \wedge \mathrm{d}\phi + \left(-\frac{\left(a^4 q - qr^4 - \left(a^4 q + a^2 qr^2\right)\sin(\theta)^2\right)\sqrt{\cos(\theta)+1}\sqrt{-\cos(\theta)+1}}{a^4 \cos(\theta)^4 + 2\,a^2 r^2 \cos(\theta)^2 + r^4}\right) \mathrm{d}\theta \wedge \mathrm{d}\phi$$

Maxwell equations

Let us check that F obeys the two (source-free) Maxwell equations:

[18]:
```
F.exterior_derivative().display()
```

[18]:

$\mathrm{d}F = 0$

[19]:
```
star_F.exterior_derivative().display()
```

[19]:

$\mathrm{d} \star F = 0$

Levi-Civita Connection

The Levi-Civita connection ∇ associated with g:

[20]:
```
nabla = g.connection()
print(nabla)
```

Levi-Civita connection nabla_g associated with the Lorentzian metric g on the 4-dimensional Lorentzian manifold M

Let us verify that the covariant derivative of g with respect to ∇ vanishes identically:

[21]:
```
nabla(g) == 0
```

[21]:

True

Another view of the above property:

[22]:
```
nabla(g).display()
```

[22]:

$$\nabla_g g = 0$$

The nonzero Christoffel symbols (skipping those that can be deduced by symmetry of the last two indices):

[23]:
```
g.christoffel_symbols_display()
```

[23]:

$$\Gamma^{t}{}_{t\,r} = \frac{a^4m+a^2q^2r+q^2r^3-mr^4-\left(a^4m+a^2mr^2\right)\sin(\theta)^2}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos(\theta)^2}$$

$$\Gamma^{t}{}_{t\,\theta} = \frac{\left(a^2q^2-2\,a^2mr\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}$$

$$\Gamma^{t}{}_{r\,\phi} = -\frac{a^3q^2r-a^3mr^2+2\,aq^2r^3-3\,amr^4-\left(a^5m+a^3q^2r-a^3mr^2\right)\cos(\theta)^4+\left(a^5m-2\,aq^2r^3+3\,amr^4\right)\cos(\theta)^2}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos(\theta)^2}$$

$$\Gamma^{t}{}_{\theta\,\phi} = \frac{\left(a^5q^2-2\,a^5mr\right)\cos(\theta)\sin(\theta)^5-\left(a^5q^2-2\,a^5mr+a^3q^2r^2-2\,a^3mr^3\right)\cos(\theta)\sin(\theta)^3}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{r}{}_{t\,t} = \frac{mr^4-\left(2\,m^2+q^2\right)r^3+\left(a^2m+3\,mq^2\right)r^2-\left(a^4m+a^2mq^2-2\,a^2m^2r+a^2mr^2\right)\cos(\theta)^2-\left(a^2q^2+q^4\right)r}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{r}{}_{t\,\phi} = -\frac{\left(amr^4-\left(2\,am^2+aq^2\right)r^3+\left(a^3m+3\,amq^2\right)r^2-\left(a^5m+a^3mq^2-2\,a^3m^2r+a^3mr^2\right)\cos(\theta)^2-\left(a^3q^2+aq^4\right)r\right)\sin(\theta)^2}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{r}{}_{r\,r} = -\frac{a^2m+q^2r-mr^2-\left(a^2m-a^2r\right)\sin(\theta)^2}{2\,mr^3-r^4-(a^2+q^2)r^2-(a^4+a^2q^2-2\,a^2mr+a^2r^2)\cos(\theta)^2}$$

$$\Gamma^{r}{}_{r\,\theta} = -\frac{a^2\cos(\theta)\sin(\theta)}{a^2\cos(\theta)^2+r^2}$$

$$\Gamma^{r}{}_{\theta\,\theta} = \frac{2\,mr^2-r^3-\left(a^2+q^2\right)r}{a^2\cos(\theta)^2+r^2}$$

$$\Gamma^{r}{}_{\phi\,\phi} = \frac{\left(a^2mr^4-\left(2\,a^2m^2+a^2q^2\right)r^3+\left(a^4m+3\,a^2mq^2\right)r^2-\left(a^6m+a^4mq^2-2\,a^4m^2r+a^4mr^2\right)\cos(\theta)^2-\left(a^4q^2+a^2q^4\right)r\right)\sin(\theta)^4+\left(2\,mr^6-r^7-\cdots\right.}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos\cdots}$$

$$\Gamma^{\theta}{}_{t\,t} = \frac{\left(a^2q^2-2\,a^2mr\right)\cos(\theta)\sin(\theta)}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{\theta}{}_{t\,\phi} = -\frac{\left(a^3q^2-2\,a^3mr+aq^2r^2-2\,amr^3\right)\cos(\theta)\sin(\theta)}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{\theta}{}_{r\,r} = -\frac{a^2\cos(\theta)\sin(\theta)}{2\,mr^3-r^4-(a^2+q^2)r^2-(a^4+a^2q^2-2\,a^2mr+a^2r^2)\cos(\theta)^2}$$

$$\Gamma^{\theta}{}_{r\,\theta} = \frac{r}{a^2\cos(\theta)^2+r^2}$$

$$\Gamma^{\theta}{}_{\theta\,\theta} = -\frac{a^2\cos(\theta)\sin(\theta)}{a^2\cos(\theta)^2+r^2}$$

$$\Gamma^{\theta}{}_{\phi\,\phi} = -\frac{\left(\left(a^6+a^4q^2-2\,a^4mr+a^4r^2\right)\cos(\theta)^5-2\,\left(2\,a^2mr^3-a^2r^4-\left(a^4+a^2q^2\right)r^2\right)\cos(\theta)^3-\left(a^4q^2-2\,a^4mr+2\,a^2q^2r^2-4\,a^2mr^3-a^2r^4-r^6\right)\cos\cdots\right.\right.}{a^6\cos(\theta)^6+3\,a^4r^2\cos(\theta)^4+3\,a^2r^4\cos(\theta)^2+r^6}$$

$$\Gamma^{\phi}{}_{t\,r} = \frac{a^3m\cos(\theta)^2+aq^2r-amr^2}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos(\theta)^2}$$

$$\Gamma^{\phi}{}_{t\,\theta} = \frac{\left(aq^2-2\,amr\right)\cos(\theta)}{\left(a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4\right)\sin(\theta)}$$

$$\Gamma^{\phi}{}_{r\,\phi} = -\frac{a^2q^2r-a^2mr^2+q^2r^3-2\,mr^4+r^5-\left(a^4m-a^4r\right)\cos(\theta)^4+\left(a^4m-a^2mr^2+2\,a^2r^3\right)\cos(\theta)^2}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos(\theta)^2}$$

$$\Gamma^{\phi}{}_{\theta\,\phi} = \frac{a^4\cos(\theta)\sin(\theta)^4-\left(2\,a^4+a^2q^2-2\,a^2mr+2\,a^2r^2\right)\cos(\theta)\sin(\theta)^2+\left(a^4+2\,a^2r^2+r^4\right)\cos(\theta)}{\left(a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4\right)\sin(\theta)}$$

Killing vectors

The default vector frame on the spacetime manifold is the coordinate basis associated with Boyer-Lindquist coordinates:

[24]:
```
M.default_frame() is BL.frame()
```

[24]:

True

[25]:
```
BL.frame()
```

[25]: $\left(\mathcal{M}, \left(\frac{\partial}{\partial t}, \frac{\partial}{\partial r}, \frac{\partial}{\partial \theta}, \frac{\partial}{\partial \phi}\right)\right)$

Let us consider the first vector field of this frame:

[26]:
```
xi = BL.frame()[0] ; xi
```

[26]: $\frac{\partial}{\partial t}$

[27]:
```
print(xi)
```

```
Vector field d/dt on the 4-dimensional Lorentzian manifold M
```

The 1-form associated to it by metric duality is

[28]:
```
xi_form = xi.down(g)
xi_form.display()
```

[28]: $\left(-\frac{q^2-2\,mr}{a^2\cos(\theta)^2+r^2}-1\right)\mathrm{d}t+\left(\frac{\left(q^2-2\,mr\right)a\sin(\theta)^2}{a^2\cos(\theta)^2+r^2}\right)\mathrm{d}\phi$

Its covariant derivative is

[29]:
```
nab_xi = nabla(xi_form)
print(nab_xi)
nab_xi.display()
```

```
Tensor field of type (0,2) on the 4-dimensional Lorentzian manifold M
```

[29]: $$\left(\frac{mr^4-\left(2\,m^2+q^2\right)r^3+\left(a^2m+3\,mq^2\right)r^2-\left(a^4m+a^2mq^2-2\,a^2m^2r+a^2mr^2\right)\cos(\theta)^2-\left(a^2q^2+q^4\right)r}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos(\theta)^2}\right)\mathrm{d}t\otimes\mathrm{d}r+\left(-\frac{\left(a^2q^2-2\,a^2mr\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}t\otimes\mathrm{d}\theta+\left(-\frac{mr^4-\left(2\,m^2+q^2\right)r^3+\left(a^2m+3\,mq^2\right)r^2-\left(a^4m+a^2mq^2-2\,a^2m^2r+a^2mr^2\right)\cos(\theta)^2-\left(a^2q^2+q^4 \cdots\right.}{2\,mr^5-r^6-(a^2+q^2)r^4-(a^6+a^4q^2-2\,a^4mr+a^4r^2)\cos(\theta)^4+2\,(2\,a^2mr^3-a^2r^4-(a^4+a^2q^2)r^2)\cos \cdots}\right.\mathrm{d}t+\left(-\frac{a^3m\cos(\theta)^4-aq^2r+amr^2-\left(a^3m-aq^2r+amr^2\right)\cos(\theta)^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}r\otimes\mathrm{d}\phi+\left(\frac{\left(a^2q^2-2\,a^2mr\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\theta\otimes\mathrm{d}t+\left(-\frac{\left(a^3q^2-2\,a^3mr+aq^2r^2-2\,amr^3\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\theta\otimes\mathrm{d}\phi+\left(\frac{a^3m\cos(\theta)^4-aq^2r+amr^2-\left(a^3m-aq^2r+amr^2\right)\cos(\theta)^2}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\phi\otimes\mathrm{d}r+\left(\frac{\left(a^3q^2-2\,a^3mr+aq^2r^2-2\,amr^3\right)\cos(\theta)\sin(\theta)}{a^4\cos(\theta)^4+2\,a^2r^2\cos(\theta)^2+r^4}\right)\mathrm{d}\phi\otimes\mathrm{d}\theta$$

Let us check that the vector field $\xi = \frac{\partial}{\partial t}$ obeys Killing equation:

[30]:
```
nab_xi.symmetrize() == 0
```

[30]: True

Similarly, let us check that $\chi := \frac{\partial}{\partial \phi}$ is a Killing vector:

[31]:
```
chi = BL.frame()[3] ; chi
```

[31]: $\frac{\partial}{\partial \phi}$

[32]:
```
nabla(chi.down(g)).symmetrize() == 0
```

[32]: True

Another way to check that ξ and χ are Killing vectors is the vanishing of the Lie derivative of the metric tensor along them:

[33]:
```
g.lie_derivative(xi) == 0
```

[33]: True

[34]:
```
g.lie_derivative(chi) == 0
```

[34]: True

Curvature

The Ricci tensor associated with g:

[35]:
```
Ric = g.ricci()
print(Ric)
```

```
Field of symmetric bilinear forms Ric(g) on the 4-dimensional Lorentzian
manifold M
```

[36]:
```
Ric.display()
```

[36]:
$$\mathrm{Ric}\left(g\right) = \left(-\frac{a^{2} q^{2} \cos\left(\theta\right)^{2} - 2 \, a^{2} q^{2} - q^{4} + 2 \, m q^{2} r - q^{2} r^{2}}{a^{6} \cos\left(\theta\right)^{6} + 3 \, a^{4} r^{2} \cos\left(\theta\right)^{4} + 3 \, a^{2} r^{4} \cos\left(\theta\right)^{2} + r^{6}} \right) \mathrm{d} t\otimes \mathrm{d} t + \left(\frac{\left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r + 2 \, a^{3} q^{2} r^{2}\right) \sin\left(\theta\right)^{4} - \left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r}{a^{8} \cos\left(\theta\right)^{8} + 4 \, a^{6} r^{2} \cos\left(\theta\right)^{6} + 6 \, a^{4} r^{4} \cos\left(\theta\right)^{4} +}$$

$$\mathrm{d} \phi + \left(\frac{q^{2}}{2 \, m r^{3} - r^{4} - {\left(a^{2} + q^{2}\right)} r^{2} - {\left(a^{4} + a^{2} q^{2} - 2 \, a^{2} m r + a^{2} r^{2}\right)} \cos\left(\theta\right)^{2}} \right) \mathrm{d} r\otimes \mathrm{d} r + \left(\frac{q^{2}}{a^{2} \cos\left(\theta\right)^{2} + r^{2}} \right) \mathrm{d} \theta\otimes \mathrm{d} \theta +$$

$$\left(\frac{\left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r + 2 \, a^{3} q^{2} r^{2}\right) \sin\left(\theta\right)^{4} - \left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r - 2 \, a m q^{2} r^{3} + 2 \, a q^{2} r^{4} + \left(4 \, a^{3} q^{2} + a q^{4}\right) r^{2}\right) \sin\left(\theta\right)^{2}}{a^{8} \cos\left(\theta\right)^{8} + 4 \, a^{6} r^{2} \cos\left(\theta\right)^{6} + 6 \, a^{4} r^{4} \cos\left(\theta\right)^{4} + 4 \, a^{2} r^{6} \cos\left(\theta\right)^{2} + r^{8}} \right) \mathrm{d} \phi\otimes$$

$$\mathrm{d} t + \left(-\frac{\left(a^{6} q^{2} + a^{4} q^{4} - 2 \, a^{4} m q^{2} r + a^{4} q^{2} r^{2}\right) \sin\left(\theta\right)^{6} - \left(a^{4} q^{4} - 2 \, a^{4} m q^{2} r + a^{2} q^{4} r^{2} - 2 \, a^{2} m q^{2} r^{3}\right) \sin\left(\theta\right)^{4} - \left(a^{6} q^{2} + 3 \, a^{4} q^{2} r^{2} + 3 \, a^{2} q^{2} r^{4} + q^{2} r^{6}\right) \sin\left(\theta\right)^{2}}{a^{8} \cos\left(\theta\right)^{8} + 4 \, a^{6} r^{2} \cos\left(\theta\right)^{6} + 6 \, a^{4} r^{4} \cos\left(\theta\right)^{4} + 4 \, a^{2} r^{6} \cos\left(\theta\right)^{2} + r^{8}} \right) \mathrm{d} \phi$$
$$\mathrm{d} \phi$$

[37]:
```
Ric[:]
```

[37]:
$$\left(\begin{array}{rr} -\frac{a^{2} q^{2} \cos\left(\theta\right)^{2} - 2 \, a^{2} q^{2} - q^{4} + 2 \, m q^{2} r - q^{2} r^{2}}{a^{6} \cos\left(\theta\right)^{6} + 3 \, a^{4} r^{2} \cos\left(\theta\right)^{4} + 3 \, a^{2} r^{4} \cos\left(\theta\right)^{2} + r^{6}} & \\ 0 & \frac{}{2 \, m r^{3} - r^{4} - {\left(a^{2} + q^{2}\right)} r^{2} -} \\ 0 & \\ \frac{\left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r + 2 \, a^{3} q^{2} r^{2}\right) \sin\left(\theta\right)^{4} - \left(2 \, a^{5} q^{2} + a^{3} q^{4} - 2 \, a^{3} m q^{2} r - 2 \, a m q^{2} r^{3} + 2 \, a q^{2} r^{4} + \left(4 \, a^{3} q^{2} + a q^{4}\right) r^{2}\right) \sin\left(\theta\right)^{2}}{a^{8} \cos\left(\theta\right)^{8} + 4 \, a^{6} r^{2} \cos\left(\theta\right)^{6} + 6 \, a^{4} r^{4} \cos\left(\theta\right)^{4} + 4 \, a^{2} r^{6} \cos\left(\theta\right)^{2} + r^{8}} & \end{array}\right.$$

Let us check that in the Kerr case, i.e. when $q = 0$, the Ricci tensor is zero:

[38]:
```
Ric_Kerr = Ric.copy()
Ric_Kerr.apply_map(lambda f: f.subs({q: 0}))
Ric_Kerr[:]
```

[38]:

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

The Riemann curvature tensor associated with g:

[39]:
```
R = g.riemann()
print(R)
```

Tensor field Riem(g) of type (1,3) on the 4-dimensional Lorentzian manifold M

The component $R^0{}_{101}$ of the Riemann tensor is

[40]:
```
R[0,1,0,1]
```

[40]:

$$\frac{4\,a^2q^2r^2-3\,a^2mr^3+3\,q^2r^4-2\,mr^5+\left(a^4q^2-3\,a^4mr\right)\cos(\theta)^4-\left(2\,a^4q^2-9\,a^4mr+2\,a^2q^2r^2-7\,a^2mr^3\right)\cos(\theta)^2}{2\,mr^7-r^8-(a^2+q^2)r^6-(a^8+a^6q^2-2\,a^6mr+a^6r^2)\cos(\theta)^6+3\,(2\,a^4mr^3-a^4r^4-(a^6+a^4q^2)r^2)\cos(\theta)^4+3\,(2\,a^2mr^5-a^2r^6-(a^4+a^2q^2)r^4)\cos(\theta)^2}$$

The expression in the uncharged limit (Kerr spacetime) is

[41]:
```
R[0,1,0,1].expr().subs(q=0).simplify_rational()
```

[41]:

$$\frac{3\,a^4mr\cos(\theta)^4+3\,a^2mr^3+2\,mr^5-\left(9\,a^4mr+7\,a^2mr^3\right)\cos(\theta)^2}{a^2r^6-2\,mr^7+r^8+(a^8-2\,a^6mr+a^6r^2)\cos(\theta)^6+3\,(a^6r^2-2\,a^4mr^3+a^4r^4)\cos(\theta)^4+3\,(a^4r^4-2\,a^2mr^5+a^2r^6)\cos(\theta)^2}$$

while in the non-rotating limit (Reissner-Nordström spacetime), it is

[42]:
```
R[0,1,0,1].expr().subs(a=0).simplify_rational()
```

[42]:

$$-\frac{3\,q^2-2\,mr}{q^2r^2-2\,mr^3+r^4}$$

In the Schwarzschild limit, it reduces to

[43]:
```
R[0,1,0,1].expr().subs(a=0, q=0).simplify_rational()
```

[43]:

$$-\frac{2\,m}{2\,mr^2-r^3}$$

Obviously, it vanishes in the flat space limit:

[44]:
```
R[0,1,0,1].expr().subs(m=0, a=0, q=0)
```

[44]:

$$0$$

Bianchi identity

Let us check the Bianchi identity $\nabla_p R^i{}_{jkl} + \nabla_k R^i{}_{jlp} + \nabla_l R^i{}_{jpk} = 0$:

[45]:
```
DR = nabla(R)    # long (takes a while)
print(DR)
```

```
Tensor field nabla_g(Riem(g)) of type (1,4) on the 4-dimensional Lorentzian
manifold M
```

[46]:
```
for i in M.irange():
    for j in M.irange():
        for k in M.irange():
            for l in M.irange():
                for p in M.irange():
                    print(DR[i,j,k,l,p] + DR[i,j,l,p,k] + DR[i,j,p,k,l], end='␣
↪')
```

```
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
```

If the last sign in the Bianchi identity is changed to minus, the identity does no longer hold:

[47]:
```
DR[0,1,2,3,1] + DR[0,1,3,1,2] + DR[0,1,1,2,3] # should be zero (Bianchi␣
↪identity)
```

[47]: 0

[48]:
```
DR[0,1,2,3,1] + DR[0,1,3,1,2] - DR[0,1,1,2,3] # note the change of the second +␣
↪to -
```

[48]:

$$-\frac{4\left(\left(a^5q^2-6\,a^5mr+a^3q^2r^2-6\,a^3mr^3\right)\cos(\theta)^3-\left(5\,a^3q^2r^2-6\,a^3mr^3+5\,aq^2r^4-6\,amr^5\right)\cos(\theta)\right)\sin(\theta)}{2\,mr^7-r^8-(a^2+q^2)r^6-(a^8+a^6q^2-2\,a^6mr+a^6r^2)\cos(\theta)^6+3\,(2\,a^4mr^3-a^4r^4-(a^6+a^4q^2)r^2)\cos(\theta)^4+3\,(2\,a^2mr^5-a^2r^6-(a^4+a^2q^2)r^4)\cos(\theta)^2}$$

1.1.1 Ricci scalar

The Ricci scalar $R = g^{ab}R_{ab}$ of the Kerr-Newman spacetime vanishes identically:

[49]:
```
g.ricci_scalar().display()
```

[49]:

$$\begin{array}{llcl} \mathrm{r}\left(g\right): & \mathcal{M} & \longrightarrow & \mathbb{R} \\ & (t,r,\theta,\phi) & \longmapsto & 0 \end{array}$$

Einstein equation

The Einstein tensor is

[50]:
```
G = Ric - 1/2*g.ricci_scalar()*g
print(G)
```

Field of symmetric bilinear forms Ric(g)-unnamed metric on the 4-dimensional Lorentzian manifold M

Since the Ricci scalar is zero, the Einstein tensor reduces to the Ricci tensor:

[51]:
```
G == Ric
```

[51]: True

The invariant $F_{ab}F^{ab}$ of the electromagnetic field:

[52]:
```
Fuu = F.up(g)
F2 = F['_ab']*Fuu['^ab'] ; print(F2)
```

Scalar field on the 4-dimensional Lorentzian manifold M

[53]:
```
F2.display()
```

[53]:

$$\begin{array}{lcl} \mathcal{M} & \longrightarrow & \mathbb{R} \\ (t,r,\theta,\phi) & \longmapsto & -\frac{2\left(a^4q^2\cos(\theta)^4-6\,a^2q^2r^2\cos(\theta)^2+q^2r^4\right)}{a^8\cos(\theta)^8+4\,a^6r^2\cos(\theta)^6+6\,a^4r^4\cos(\theta)^4+4\,a^2r^6\cos(\theta)^2+r^8} \end{array}$$

The energy-momentum tensor of the electromagnetic field:

[54]:
```
Fud = F.up(g,0)
T = 1/(4*pi)*( F['_k.']*Fud['^k_.'] - 1/4*F2 * g ); print(T)
```

Tensor field of type (0,2) on the 4-dimensional Lorentzian manifold M

[55]:
```
T[:]
```

[55]:

$$\left(\begin{array}{ccc} -\frac{a^2q^2\cos(\theta)^2-2\,a^2q^2-q^4+2\,mq^2r-q^2r^2}{8\left(\pi a^6\cos(\theta)^6+3\,\pi a^4r^2\cos(\theta)^4+3\,\pi a^2r^4\cos(\theta)^2+\pi r^6\right)} & 0 & \cdots \\ 0 & \frac{q^2}{8\left(2\,\pi mr^3-\pi r^4-(\pi a^2+\pi q^2)r^2-(\pi a^4+\pi a^2q^2-2\,\pi a^2mr+\pi a^2r^2)\cos(\theta)^2\right)} & \cdots \\ 0 & 0 & \frac{\cdots}{8\left(\pi a^2 \cdots\right.} \\ -\frac{\left(2\,a^3q^2+aq^4-2\,amq^2r+2\,aq^2r^2\right)\sin(\theta)^2}{8\left(\pi a^6\cos(\theta)^6+3\,\pi a^4r^2\cos(\theta)^4+3\,\pi a^2r^4\cos(\theta)^2+\pi r^6\right)} & 0 & \cdots \end{array}\right.$$

Check of the Einstein equation:

[56]:
```
G == 8*pi*T
```

[56]: True

1.1.2 Kretschmann scalar

The tensor $R^\flat$, of components $R_{abcd} = g_{am} R^m{}_{bcd}$:

[57]:
```
dR = R.down(g)
print(dR)
```

Tensor field of type (0,4) on the 4-dimensional Lorentzian manifold M

The tensor $R^\sharp$, of components $R^{abcd} = g^{bp} g^{cq} g^{dr} R^a{}_{pqr}$:

[58]:
```
uR = R.up(g)
print(uR)
```

Tensor field of type (4,0) on the 4-dimensional Lorentzian manifold M

The Kretschmann scalar $K := R^{abcd} R_{abcd}$:

[59]:
```
Kr_scalar = uR['^ijkl']*dR['_ijkl']
Kr_scalar.display()
```

[59]:

$$\begin{array}{llcl} & \mathcal{M} & \longrightarrow & \mathbb{R} \\ & (t,r,\theta,\phi) & \longmapsto & -\frac{8\left(6\,m^2r^8-12\left(m^3+mq^2\right)r^7+\left(6\,a^2m^2+30\,m^2q^2+7\,q^4\right)r^6-6\left(a^8m^2+a^6m^2q^2-2\,a^6m^3r+a^6m^2r^2\right)\cos(\theta)^6-2\left(6\,a^2mq^2+\cdots\right.}{2\,mr^{13}-r^{14}-(a^2+q^2)r^{12}-(a^{14}+a^{12}q^2-2\,a^{12}mr+a^{12}r^2)\cos\cdots} \end{array}$$

A variant of this expression can be obtained by invoking the factor() method on the coordinate function representing the scalar field in the manifold's default chart:

[60]:
```
Kr = Kr_scalar.coord_function()
Kr.factor()
```

[60]:

$$-\frac{8\left(6\,a^6m^2\cos(\theta)^6-7\,a^4q^4\cos(\theta)^4+60\,a^4mq^2r\cos(\theta)^4-90\,a^4m^2r^2\cos(\theta)^4+34\,a^2q^4r^2\cos(\theta)^2-120\,a^2mq^2r^3\cos(\theta)^2+90\,a^2m^2r^4\cos(\theta)^2-7\,q^4r^4+12\cdots\right.}{\left(a^2\cos(\theta)^2+r^2\right)^6}$$

As a check, we can compare Kr to the formula given by R. Conn Henry, Astrophys. J. 535, 350 (2000):

[61]:
```
Kr == 8/(r^2+(a*cos(th))^2)^6 *(
          6*m^2*(r^6 - 15*r^4*(a*cos(th))^2 + 15*r^2*(a*cos(th))^4 -␣
↪(a*cos(th))^6)
        - 12*m*q^2*r*(r^4 - 10*(a*r*cos(th))^2 + 5*(a*cos(th))^4)
        + q^4*(7*r^4 - 34*(a*r*cos(th))^2 + 7*(a*cos(th))^4) )
```

[61]: True

The Schwarzschild value of the Kretschmann scalar is recovered by setting $a = 0$ and $q = 0$:

[62]:
```
Kr.expr().subs(a=0, q=0)
```

[62]: $\frac{48\,m^2}{r^6}$

Let us plot the Kretschmann scalar for $m = 1$, $a = 0.9$ and $q = 0.5$:

[63]:
```
K1 = Kr.expr().subs(m=1, a=0.9, q=0.5)
plot3d(K1, (r,1,3), (th, 0, pi), axes_labels=['r', 'theta', 'Kr'])
```

[63]: Graphics3d Object

[64]:
```
print("Total elapsed time: {} s".format(time.perf_counter() - comput_time0))
```

Total elapsed time: 913.9951457749994 s